{ TC "'APPENDIX A: EES RESISTOR OPTIMIZATION PROGRAM':" \l 1 \n }APPENDIX

A

EES RESISTOR OPTIMIZATION PROGRAM

A.1 EES Photovoltaic Array Procedure{ TC "A.1 EES Photovoltaic Array Procedure" \l 2 }

```
"THIS PROCEDURE DETERMINES THE OPERATING CURRENT OF A PV ARRAY GIVEN THE OPERATING
VOLTAGE AND PV MODULE CHARACTERISTIC PARAMETERS BY THE FOUR PARAMETER METHOD AS
DESCRIBED BY DUFFIE AND BECKMAN (1991)"

Procedure
Photovoltaic(area,T_c_ref,G_ref,I_sc_ref,V_oc_ref,I_mp_ref,V_mp_ref,mu_I_sc,mu_V_oc,NSC,NS,NP,T,G,V:I)
epsilon:=1.15 "material bandgap energy for silicon"
R_sh = 100000 "shunt resistance taken to be infinite"
beta=3.0

{Module Parameters @ the reference operating condition}
I_L_ref:=I_sc_ref
a_ref:=(mu_V_oc*T_c_ref-V_oc_ref+epsilon*NSC)/((mu_I_sc*T_c_ref/I_L_ref)-beta)
I_o_ref:=I_L_ref*exp(-V_oc_ref/a_ref)
R_s_ref:=(a_ref*ln(1-I_mp_ref/I_L_ref)-V_mp_ref+V_oc_ref)/I_mp_ref

{Array Parameters @ the reference operating condition}
I_L_ref=NP*I_L_ref
I_o_ref=NP*I_o_ref
a_ref=NS*a_ref
R_s_ref=(NS/NP)*R_s_ref

{Array Parameters @ the desire operating condition}
I_L:=G/G_ref*(I_L_ref+mu_I_sc*NP*(T-T_c_ref))
IO:=I_o_ref*(T/T_c_ref)^beta*exp((epsilon*NS*NSC/a_ref)*(1-T_c_ref/T))
a:=a_ref*T/T_c_ref
R_s:=R_s_ref

{Starting Newton-Raphsoson-Method}
 I_0=0
 I_1:=G/G_ref*(I_mp_ref+mu_I_sc*(T-T_c_ref)){I_mp_ref}
Repeat
    I_0=I_1
    f_I_0:=I_0-I_L+IO*(exp((V+I_0*R_s)/a)-1)+((V+I_0*R_s)/R_sh)
    df_I_0:=1+(IO*R_s/a)*(exp((V+I_0*R_s)/a){-1})+V*R_s/R_sh
    I_1:=I_0-(f_I_0/df_I_0)
 Until  (abs(I_1-I_0)<1e-6)
 I=I_1
```

2

End

A.2 EES PV-SDHW System Resistor Optimization Program{ TC "A.2 EES PV-SDHW System Resistor Optimization Program" \l 2 }

"
RESISTOR SELECTION FOR PV WATER HEATING SYSTEMS BASED IN AN IRRADIANCE-BASED CONTROL SCHEME

THIS PROGRAM ESTIMATES THE OPTIMAL VALUES FOR THE RESISTORS USED IN PV WATER HEATING SYSTEMS. THE PROGRAM CAN BE CONFIGURED TO DETERMINE THE SIX BEST RESISTORS FOR USE IN A TWO-TANK PV WATER
HEATING SYSTEM (WHERE SIX RESISTORS IN THE PREHEAT TANK ARE CONFIGURED IN SIX PARALLEL COMBINATIONS AS IRRADIANCE VARIES) OR THE THREE BEST RESISTORS FOR USE IN A ONE-TANK PV WATER HEATING SYSTEM (WHERE THREE RESISTORS IN THE BOTTOM OF THE WATER TANK ARE CONFIGURED IN ALL SEVEN PARALLEL COMBINATIONS
AS IRRADIANCE VARIES).

THE PROGRAM WORKS BY FINDING THE COMBINATION OF RESISTORS (EITHER SIX OR THREE) WHICH MINIMIZES THE SUM OF THE SQUARES OF THE ERRORS BETWEEN THE MAXIMUM POWER AND POWER OF THE SIX- OR THREE-RESISTOR SET FOR 20 IRRADIANCE POINTS BETWEEN 50 AND 1000 W/M2.

PV ARRAYS ARE MODELED IN THIS PROGRAM USING THE FOUR-PARAMETER, SINGLE DIODE EQUIVALENT CIRCUIT MODEL (DUFFIE AND BECKMAN, 1991). THE MODEL IS ENCODED IN THE EES LIBRARY FILE 'PVARAY.LIB'.

TO USE THE MODEL:

(THESE INSTRUCTIONS APPLY TO THE SIX-RESISTOR CASE)

1. FILL IN THE SECTION CALLED 'PV module characteristics' WITH THE PARAMETERS DESCRIBING THE DESIRED ARRAY (IT IS SET UP NOW FOR THE SIEMANS M55 MODULE AS DETERMINED BY NIST,1995). INCLUDE THE NUMBER OF MODULES IN SERIES AND IN PARALLEL WITHIN THE ARRAY.

2. THIS STEP SETS UP THE LOOKUP TABLE WITH DATA ON MAXIMUM POWER AS A FUNCTION OF IRRADIANCE FOR THE ARRAY. SET n (NUMBER OF RESISTANCES TO CONSIDER) AND irrad# (NUMBER OF IRRADIANCE POINTS AT WHICH ERROR IS MINIMIZED) TO 1 IN THE WORKSHEET. COMMENT OUT THE SECTION OF THE PROGRAM NEAR THE BOTTOM AS SHOWN IN THE PROGRAM TEXT. CHOOSE MIN/MAX TABLE AND CHOOSE TO MAXIMIZE P[1,1] AS A FUNCTION OF R[1]. WHEN THE MAXIMIZATION IS FINISHED, COPY THE COLUMN P[1,1] IN THE PARAMETRIC TABLE TO THE LOOKUP TABLE UNDER THE COLUMN 'Pmax at 10C'.

3. NOW GO BACK TO THE WORKSHEET AND CHANGE n BACK TO 6 AND irrad# BACK TO 20. UNCOMMENT THE SECTION COMMENTED OUT IN STEP 2. UNCOMMENT THE STATEMENTS AT THE END OF THE PROGRAM FIXING ALL THE RESISTANCES TO A UNIFORM VALUE. CHOOSE SOLVE THEN UPDATE GUESS VALUES. THIS WILL ESTABLISH REASONABLE GUESS VALUES FOR THE MANY PROGRAM VARIABLES. NOW COMMENT THE RESISTOR ASSIGNMENTS AGAIN. CHOOSE MIN/MAX AND CHOOSE TO MINIMIZE sumofsquares AS A FUNCTION OF resistor[1] THROUGH resistor[6]. THIS MINIMIZATION WILL TAKE A FEW MINUTES.

4. WHEN THE MINIMIZATION IS COMPLETE, THE ARRAYS TABLE CONTAINS THE RESULTS. THE resistor[k] COLUMN SHOWS THE SIX RESISTORS WHICH WERE SELECTED. THE switch[j] COLUMN SHOWS THE IRRADIANCE LEVEL SWITCHPOINTS (TO WITHIN 25 W/M2) TO BE USED WITH THESE RESISTORS FOR MAXIUMUM OVERALL POWER OUTPUT.

NOTES:
IF IT IS DESIRED TO FIND A SINGLE OPTIMAL RESISTOR VALUE TO BE USED IN SIX PARALLEL COMBINATIONS, UNCOMMENT THE APPROPRIATE LINE IN THE PROGRAM WHICH STATES 'resistor[1]=resistor[2]; resistor[2]=resistor[3]; ...etc'. IN THIS CASE YOU WOULD MINIMIZE sumofsquares AS A FUNCTION ONLY OF resistor[1] IN STEP 3 ABOVE.

FOR THE CASE OF FINDING THE OPTIMAL THREE RESISTORS TO BE USED IN SEVEN COMBINATIONS, COMMENT OUT THE APPROPRIATE SIX-RESISTOR SECTION IN THE PROGRAM AND UNCOMMENT THE THREE-RESISTOR SECTION. TO RUN THIS CASE, THE SEVENTH P[k] AND R[k] VALUES MUST BE UNCOMMENTED IN THE PROCEDURE DECLARATION AND CALL FOR 'Procedure P_actual' AND THE LAST 'IF-THEN' CASE IN THAT PROCEDURE MUST BE UNCOMMENTED. ALSO IN THAT PROCEDURE, USE THE RIGHT LINES SETTING 'r_on_1=0, etc..' SET n TO 7 RATHER THAN 6 IN STEP 3 ABOVE. USE THE APPROPRIATE INITIALIZATION SET AT THE BOTTOM OF THE PROGRAM. MAKE OTHER ADJUSTMENTS WHERE NECESSARY."

```
"This procedure determines the resistance which results in maximum power given six or seven sets of power and resistance:"
Procedure
P_actual(P_1,P_2,P_3,P_4,P_5,P_6{,P_7},R_1,R_2,R_3,R_4,R_5,R_6{,R_7}:P,Rused,r_on_1,r_on_2,r_on_3,r_on_4,r_on_5,r_on_6)
P=P_1
Rused=R_1
r_on_1=1;r_on_2=0;r_on_3=0;r_on_4=0;r_on_5=0;r_on_6=0
If (P_2>=P) Then
  P=P_2;  Rused=R_2
  r_on_1=1;r_on_2=1;r_on_3=0;r_on_4=0;r_on_5=0;r_on_6=0
 { r_on_1=0;r_on_2=1;r_on_3=0}
EndIf
If (P_3>=P) Then
   P=P_3; Rused=R_3
   r_on_1=1;r_on_2=1;r_on_3=1;r_on_4=0;r_on_5=0;r_on_6=0
 { r_on_1=1;r_on_2=1;r_on_3=0}
EndIf
If (P_4>=P) Then
  P=P_4;  Rused=R_4
  r_on_1=1;r_on_2=1;r_on_3=1;r_on_4=1;r_on_5=0;r_on_6=0
   {r_on_1=0;r_on_2=0;r_on_3=1}
EndIf
If (P_5>=P) Then
  P=P_5;  Rused=R_5
  r_on_1=1;r_on_2=1;r_on_3=1;r_on_4=1;r_on_5=1;r_on_6=0
   {r_on_1=1;r_on_2=0;r_on_3=1}
EndIf
If (P_6>=P) Then
  P=P_6;  Rused=R_6
  r_on_1=1;r_on_2=1;r_on_3=1;r_on_4=1;r_on_5=1;r_on_6=1
 { r_on_1=0;r_on_2=1;r_on_3=1}
EndIf
{If (P_7>=P) Then
```

```
  P=P_7;  Rused=R_7
  r_on_1=1;r_on_2=1;r_on_3=1
Endif}
end


"PV module characteristics:"
A_m=0.427 "module area"
T_c_ref=45.15+273.15 "cell temperature at reference conditions"
G_T_ref=904.8 "irradiance at reference conditions"
V_oc_ref_mod=199.51/10 "open circuit voltage of module at reference conditions"
I_sc_ref_mod=12.225/4 "short circuit current of module at reference conditions"
V_mp_ref=152.48/10 "maximum power voltage of module at reference conditions"
I_mp_ref=11.125/4 "maximum power current of module at reference conditions"
mu_V_oc=-0.7561/10 "temperature coefficient of open circuit voltage for module"
mu_I_sc=(7.27e-3)/4 "temperature coefficient of short circuit current for module"
N_s=36 "number of cells in series in module"
NS=10 "number of modules in series in array"
NP=3 "number of modules in parallel in array"
A_a=A_m*NS*NP "array area"

n=6 "number of overall resistances to consider"
n_elements=6 "number of resistor elements present"
irrad#=20 "number of irradiance levels"

"For each resistance (R[k]), irradiance (G_T[j]), and array temperature (T_c[j,k]) call the procedure Photovoltaic
to determine the array voltage (V[j,k]), current (I[j,k]), and power (P[j,k]):"
Duplicate k=1,n
Duplicate j=1,irrad#
Call
Photovoltaic(A_m,T_c_ref,G_T_ref,I_sc_ref_mod,V_oc_ref_mod,I_mp_ref,V_mp_ref,mu_I_sc,mu_V_oc,N_s,N
S,NP,T_c[j,k],G_T[j],V[j,k]:I[j,k])
P[j,k]=I[j,k]*V[j,k]
V[j,k]=I[j,k]*R[k]
End
End

"Determine a constant overall heat transfer coefficient (U_L) for the array based on NOCT conditions:"
taualpha\U_L=(T_c_NOCT-T_a_NOCT)/G_T_NOCT
T_c_NOCT=42
T_a_NOCT=20
G_T_NOCT=800
taualpha=0.9

"For each resistance at each irradiance (G_T[j]), calculate the cell temperature (T_c[j,k]) based on an energy
balance on the array.  Also necessary is a calculation of the array efficiency (eta_c[j,k]) for each resistance and
irradiance.:"
Duplicate k=1,n
Duplicate j=1,irrad#
T_c[j,k]=T_a+(G_T[j]*taualpha\U_L)*(1-eta_c[j,k]/taualpha)
eta_c[j,k]=(I[j,k]*V[j,k])/(A_a*G_T[j])
End
End

"Set the ambient temperature to an estimated constant  intermediate value of 10 C:"
T_a=10.0+273.15
```

```
"****************************************************************************************
*********************************"
"COMMENT OUT THE PROGRAM FROM HERE TO THE END WHEN SETTING UP THE LOOKUP TABLE
WITH
MAXIMUM POWER DATA (STEP 2)"
"****************************************************************************************
*********************************"

{
"For the case of three resistors and seven resistances, relate the three resistor values (resistor[k])  to the seven
resistance values (R[k]):"
R[1]=resistor[1]
R[2]=resistor[2]
1/R[3]=1/resistor[1]+1/resistor[2]
R[4]=resistor[3]
1/R[5]=1/resistor[1]+1/resistor[3]
1/R[6]=1/resistor[2]+1/resistor[3]
1/R[7]=1/resistor[1]+1/resistor[2]+1/resistor[3]
"If using three equal resistors, uncomment this line:"
{resistor[1]=resistor[2];resistor[2]=resistor[3]}
}


"For the case of six resistors and six resistances, relate the six resistor values (resistor[k])  to the six resistance
values (R[k]):"
1/R[1]=1/resistor[1]
1/R[2]=1/resistor[1]+1/resistor[2]
1/R[3]=1/resistor[1]+1/resistor[2]+1/resistor[3]
1/R[4]=1/resistor[1]+1/resistor[2]+1/resistor[3]+1/resistor[4]
1/R[5]=1/resistor[1]+1/resistor[2]+1/resistor[3]+1/resistor[4]+1/resistor[5]
1/R[6]=1/resistor[1]+1/resistor[2]+1/resistor[3]+1/resistor[4]+1/resistor[5]+1/resistor[6]
"If using six equal resistors, uncomment this line:"
{resistor[1]=resistor[2];resistor[2]=resistor[3];resistor[3]=resistor[4];resistor[4]=resistor[5];resistor[5]=resistor[6]
}


"Read in the irradiance levels (G_T[j])  from the lookup table:"
Duplicate j=1,irrad#
G_T[j]=lookup(j,2)
End

"For each irradiance level, send the resistance values (R[k]) along with the array power each would yield (P[j,k])
to the procedure P_actual to determine which resistance  yields the greatest power (P[j]) at that irradiance and
what the resistance is (R_used[j]).  The procedure also returns a flag for each resistor (Ron[j,k]) indicating
whether it is engaged (Ron[j,k]=1) or disengaged (Ron[j,k]=0) at that irradiance.:"
Duplicate j=1,irrad#
Call
P_actual(P[j,1],P[j,2],P[j,3],P[j,4],P[j,5],P[j,6]{,P[j,7]},R[1],R[2],R[3],R[4],R[5],R[6]{,R[7]}:P[j],R_used[j],Ron[j,1],Ro
n[j,2],Ron[j,3],Ron[j,4],Ron[j,5],Ron[j,6])
"Determine whether a resistance switch occurs between adjacent irradaince levels and mark with a flag (sw[j]):"
sw[j]=step(R_used[j-1]-R_used[j]-1e-9)
"Determine the irradiance level switchpoint (switch[j]) to the nearest 25 W/m2:"
switch[j]=sw[j]*(G_T[j]-1000/(irrad#*2))
End
```

```
R_used[0]=R_used[1]

"Define the error between the maximum power (in lookup table) and the actual power (P[j]) for each irradiance point (error[j]):"
Duplicate j=1,irrad#
error[j]=lookup(j,3)-P[j]
End

"Calculate the sum of the squares of the errors between actual and maximum power for all irradiance points:"
sumofsquares=sum(error[j]^2,j=1,irrad#)

"Determine the currents in each of the six resistors at each irradiance level (I_actual[j,k]):"
duplicate j=1,irrad#
P[j]=V_actual[j]^2/R_used[j]
duplicate k=1,n_elements
I_actual[j,k]=(V_actual[j]/resistor[k])*Ron[j,k]
end
end

"Variable initialization set for 6-resistor optimization:"
{resistor[1]=120
resistor[2]=120
resistor[3]=120
resistor[4]=120
resistor[5]=120
resistor[6]=120}

"Variable initialization set for 3-resistor optimization:"
{resistor[1]=50
resistor[2]=50
resistor[3]=50}
```